
//------------------------------------------------------------------ 
// application pour console de commande d'avion (FSX/P3D) 14/10/2016 
//------------------------------------------------------------------ 
// variables globales pour la conservation des valeurs 
int joyX; 
int joyY; 
int neutre; 
int addX; 
int addY; 
int moteur; 
int rich; 
int pas; 
unsigned long temp; 
bool boutonA; 
bool boutonB; 
bool boutonC; 
bool boutonD; 
bool boutonE; 
bool boutonF; 
bool boutonG; 
bool start; 
//---------------------------------------------------------------- 
void setup() { 
  Serial.begin(9600); 
  delay(50); 
  // initialisation potentiometres 
  neutre = 0;   // reglage pour retour à zéro du joystick (à ajuster selon) 
  addX = 0;    // reglage decalage zero du joystick (à ajuster selon) 
  addY = 0;    // reglage decalage zero du joystick (à ajuster selon) 
  temp = 0; 
  start = false; 
  initPotar(joyX, 'X', 0, -511, 512, neutre, addX); 
  initPotar(joyY, 'Y', 1, -511, 512, neutre, addY); 
  initPotar(moteur, 'M', 2, 0, 1023, 3, 0); 
  initPotar(rich, 'R', 3, 0, 1023, 3, 0); 
  initPotar(pas, 'P', 4, -511, 512, 3, 0); 
  // initialisation boutons 
  initBouton(boutonA, 'A', 8); 
  initBouton(boutonB, 'B', 9); 
  initBouton(boutonC, 'C', 10); 
  initBouton(boutonD, 'D', 11); 
  // initialisation push-boutons 
  initPush(boutonE, 5); 
  initPush(boutonF, 6); 
  initPush(boutonG, 7); 
} 
//---------------------------------------------------------------- 
void loop() { 
  delay(50); 
  // capture la valeur des potentiometres 
  potar(joyX, 'X', 0, -511, 512, neutre, addX); 
  potar(joyY, 'Y', 1, -511, 512, neutre, addY); 
  potar(moteur, 'M', 2, 0, 1023, 3, 0); 
  potar(rich, 'R', 3, 0, 1023, 3, 0); 



  potar(pas, 'P', 4, -511, 512, 3, 0); 
  // capture l'état des boutons 
  bouton(boutonA, 'A', 8); 
  bouton(boutonB, 'B', 9); 
  bouton(boutonC, 'C', 10); 
  bouton(boutonD, 'D', 11); 
  // capture l'état des push-boutons 
  push(boutonE, 'E', 5); 
  push(boutonF, 'F', 6); 
  push(boutonG, 'G', 7); 
  if (start) { 
    initPotar(joyX, 'X', 0, -511, 512, neutre, addX); 
    initPotar(joyY, 'Y', 1, -511, 512, neutre, addY); 
    initPotar(moteur, 'M', 2, 0, 1023, 3, 0); 
    initPotar(rich, 'R', 3, 0, 1023, 3, 0); 
    initPotar(pas, 'P', 4, -511, 512, 3, 0); 
    initBouton(boutonA, 'A', 8); 
    initBouton(boutonB, 'B', 9); 
    initBouton(boutonC, 'C', 10); 
    initBouton(boutonD, 'D', 11); 
    start = false; 
  } 
} 
//---------------------------------------------------------------- 
void initPotar(int &arg, char c, int canal, int q1, int q2, int ntr, int 
add) { 
  arg = map(analogRead(canal), 0, 1023, q1, q2) + add; 
  if (abs(arg) <= ntr) arg = 0; 
  Serial.print(" "); 
  Serial.print(arg); 
  Serial.print(c); 
  Serial.print("."); 
} 
//---------------------------------------------------------------- 
void potar(int &arg, char c, int canal, int q1, int q2, int ntr, int add) { 
  int val = map(analogRead(canal), 0, 1023, q1, q2) + add; 
  if (abs(val) <= ntr) val = 0; 
  if (abs(arg - val) > 4) { 
    arg = val; 
    Serial.print(" "); 
    Serial.print(arg); 
    Serial.print(c); 
    Serial.print("."); 
  } 
} 
//---------------------------------------------------------------- 
void initBouton(bool &arg, char c, int canal) { 
  pinMode(canal, INPUT_PULLUP); 
  arg = digitalRead(canal); 
  Serial.print(" "); 
  Serial.print(arg); 
  Serial.print(c); 
  Serial.print("."); 
} 



//---------------------------------------------------------------- 
void bouton(bool &arg, char c, int canal) { 
  bool val = digitalRead(canal); 
  if (val != arg) { 
    arg = val; 
    Serial.print(" "); 
    Serial.print(arg); 
    Serial.print(c); 
    Serial.print("."); 
  } 
} 
//---------------------------------------------------------------- 
void initPush(bool &arg, int canal) { 
  pinMode(canal, INPUT_PULLUP); 
  arg = true; 
} 
//---------------------------------------------------------------- 
void push(bool &arg, char c, int canal) { 
  if ((millis() - temp) < 200) return; 
  bool val = digitalRead(canal); 
  if (val != arg) { 
    arg = val; 
    if (!arg) { 
      Serial.print(" "); 
      Serial.print(1); 
      Serial.print(c); 
      Serial.print("."); 
      start = (canal == 7); 
      temp = millis(); 
    } 
  } 
} 
//--------------------------------------------- fin ---------------- 


